

# Simple Regression with Measurement Error: With and Without intercepts

```
/* noint.sas */
options linesize=79 noovp formdlim='_';
title 'STA2201s06 Simple Measurement Error Regression';
title2 'Assignment 5 Check: No Intercept';
data ipath;
  infile 'itest.dat.txt';
  input Y X1 X2;

proc calis cov vardef=n;
  var Y X1 X2;
  lineqs
    Y = gamma F + e,
    X1 = F + delta1,
    X2 = F + delta2;
  std          /* Variances (not standard deviations) */
    F = phi,
    e = psi,
    delta1 = theta1,
    delta2 = theta2;
  bounds 0.0 < phi theta1 theta2;

/* measint.sas */
options linesize=79 noovp formdlim='_';
title 'STA2201s06 Simple Measurement Error Regression with Intercept';

data ipath;
  infile 'itest.dat.txt';
  input Y X1 X2;

proc calis ucov aug vardef=n;
  /* ucov and aug together give intercept */
  title2 'Default starting values';
  var Y X1 X2;
  lineqs
    F = kappa intercept + eek,          /* E(F)=kappa */
    Y = alpha intercept + gamma F + e,
    X1 = F + delta1,
    X2 = nu intercept + F + delta2;
  std          /* Variances (not standard deviations) */
    eek = phi,                          /* Var(F)=phi */
    e = psi,
    delta1 = theta1,
    delta2 = theta2;
  bounds 0.0 < phi psi theta1 theta2;
```

```
/* What we want is
```

```
Y = alpha + gamma Kuhsee + zeta  
X1 = Kuhsee + delta1  
x2 = Nu + Kuhsee + delta2
```

```
with
```

```
E(Kuhsee)=kappa  
V(Kuhsee)=phi V(zeta)=psi V(delta1)=theta1 V(delta2)=theta2
```

Error terms have to begin with e or d, so we replace zeta with e. Latent variables must begin with F, so we use F (for Factor) in place of Kuhsee.

Using the options ucov and aug together allow the specification of models with an intercept. Ucov means fit a model using the "uncorrected" covariance matrix -- that is, without subtracting off  $\bar{x}$  from the observations. Aug means "augment" the matrix by adding a variable called "intercept," which has constant value one. Intercept terms have to be multiplied by this built-in variable "intercept" -- see above. This is a clever work-around, and not a desirable syntax for specifying the model. However, we can live with it.

But also, latent variables always have expected value zero by default in proc calis. This means that in order to make our latent independent variable F have a non-zero mean, we must explicitly add a constant to it. That's why we say:

$$F = \text{kappa intercept} + \text{eek}$$

So now  $E(F) = \text{kappa}$  and  $V(F) = V(\text{eek})$ , which I called phi. Without this, there is no kappa at all in the model, and the result is a disaster. Eeek!

```
*/
```

**No-intercept output:** Objective Function 0.0002359465

Manifest Variable Equations with Estimates

Y = 1.1964\*F + 1.0000 e  
 Std Err 0.3943 gamma  
 t Value 3.0341  
 X1 = 1.0000 F + 1.0000 delta1  
 X2 = 1.0000 F + 1.0000 delta2

Variances of Exogenous Variables

Variable	Parameter	Estimate	Standard Error	t Value
F	phi	1.22290	0.43289	2.82
e	psi	2.89575	0.73550	3.94
delta1	theta1	3.47072	0.55224	6.28
delta2	theta2	4.41848	0.63665	6.94

**With-intercept output:** Objective Function 0.0002359472

Manifest Variable Equations with Estimates

Y = 1.1965\*F + -1.0803\*Intercept + 1.0000 e  
 Std Err 0.3943 gamma 3.8872 alpha  
 t Value 3.0341 -0.2779  
 X1 = 1.0000 F + 1.0000 delta1  
 X2 = 1.0000 F + 0.00751\*Intercept + 1.0000 delta2  
 Std Err 0.2301 nu  
 t Value 0.0327

Latent Variable Equations with Estimates

F = 9.8403\*Intercept + 1.0000 eek  
 Std Err 0.1775 kappa  
 t Value 55.4427

Variances of Exogenous Variables

Variable	Parameter	Estimate	Standard Error	t Value
Intercept		1.00000		
eek	phi	1.22289	0.43289	2.82
e	psi	2.89575	0.73551	3.94
delta1	theta1	3.47082	0.55225	6.28
delta2	theta2	4.41838	0.63664	6.94